```
 10REM** DIALLER copyright MTL 1984 **
 15REM**      FOR BBC B COMPUTERS     **
 20MODE6
 30PROCheader
 40PROCdial
 50PROCcallTerminal
 60END
 70DEFPROCdialout
 80TIME=0:REPEAT:?&FE08=2:UNTIL TIME=100
 90*FX8,7
100FOR OUTPUT=1 TO DIAL
110?&FE08=66
120PRINT"*";
130REM**  SET DIAL MARK & SPACE HERE. THIS IS 400mS Mark, 500mS Space, approx

140TIME=0:REPEAT:?&FE09=0:UNTIL TIME=4
150REM**  &FE09=0 takes transmitted      data line from low to high
160?&FE08=2
170REM**  &FE08=2 sets RTS (DTR) High
180TIME=0:REPEAT:UNTIL TIME=5
190NEXT OUTPUT
200PRINT
210ENDPROC
220DEFPROCheader
230CLS
240PRINT"TYPE IN NUMBER, PRESS RETURN"
250INPUT N$
260CLS
270ENDPROC
280DEFPROCpause(n)
290TIME=0:REPEAT UNTIL TIME=n*100
300ENDPROC
310DEFPROCdial
320PRINTTAB(10,10)"D I A L L I N G"
330PROCpause(1)
340X=LEN(N$)
350P=0
360FOR P=1  TO X
370DIGIT$=MID$(N$,P,1)
380DIAL=VAL(DIGIT$)
390PRINT DIGIT$;
400IF DIAL=0 THEN DIAL=10
410PROCdialout
420PROCpause(.5)
430NEXT P
440ENDPROC
450DEFPROCcallTerminal
460PRINTTAB(10,10)"W A I T I N G   "
470PROCpause(5)
480PRINT:PRINT
490CLS
500REM**  This is a standard Terminal   Programme from here on to END
510VDU19,2,0,0,0,0,19,0,4,0,0,0,12
520*FX7,3
530*FX8,3
540*FX2,2
550A=INKEY(1): IF A =-1 THEN 600
560*FX3,7
570VDU A
580*FX3,0
590GOTO 540
600*FX2,1
610*FX3,0
620A=INKEY(1)
630IF A>31 THEN VDU A AND 127
640IF A =41 THEN GOSUB 670
650IF A=13 OR A=10 THEN VDU A
660GOTO 540
670A=10:VDU A
680A=13:VDU A
690RETURN
```

```
10 REM**        AUTO ANSWER PROGRAM      ** Copyright MTL 1984
20 REM
30 REM PLEASE NOTE!
40 REM LINES 400-410 AND 520-530 TEST
50 REM TO SEE IF YOU ARE RECEIVING A
60 REM VALID CARRIER FROM THE RS423 PORT
70 REM ?&FE08=66 WILL DISABLE THE MODEM
80 REM AND DROP THEN PHONE LINE
90 REM ?&FE08=2 WILL ENABLE THE MODEM
100 REM AND HOLD THE PHONE LINE IF YOU
110 REM ARE ON LINE AT THAT TIME
120 MODE 7
130 C1=0
140 CLS: PRINT:PRINT:PRINT
150 *FX2,2
160 CLS:C=0
170 MODE6
180 VDU19,2,0,0,0,0,19,0,4,0,0,0,12
190 *FX7,3
200 C=0: C1=C1+1
210 *FX8,3
220 *FX2,2
230 A=INKEY(1): IF A =-1 THEN 280 ELSE C=0:C1=0
240 *FX3,7
250 IF A=13 THEN GOSUB 370 ELSE VDU A
260 *FX3,0
270 GOTO 220
280 *FX2,1
290 *FX3,1
300 A=INKEY(1) :IF A=-1 THEN C=C+1 ELSE C=0:C1=0
310 IF C> 100 THEN GOTO  400
320 IF C1> 10 THEN ?&FE08=66:FOR T=1TO1000:NEXT T: GOTO 120
330 IF A>31 THEN VDU A AND 127
340 IF A =13 THEN GOSUB 370
350 IF  A=10 THEN VDU A
360 GOTO 220
370 A=10:VDU A
380 A=13:VDU A
390 RETURN
400 D=?&FE08
410 IF D AND 8 THEN ?&FE08=66:GOTO 430
420 GOTO 190
430 REM THIS IS THE PART OF THE PROGRAM
440 REM YOU CAN TAILOR TO YOUR OWN NEEDS
450 REM TO RUN A DATA BASE TYPE SYSTEM.
460 REM THE SMALL PROGRAM WHICH FOLLOWS
470 REM SENDS OUT WHATEVER IS IN THE
480 REM PRINT STATEMENTS (Lines 570-630) AND THEN GOES INTO
490 REM THE TERMINAL PROGRAM ABOVE
500 FOR T=1 TO 10000 :NEXT T:CLS:PRINT"AUTO ANSWER":PRINT
510 ?&FE08=2
520 D=?&FE08
530 IF D AND 8 THEN GOTO 520
540 FOR T=1 TO 4000:NEXT T
550  *FX 21,1
560 *FX3,7
570  FOR P=1 TO 30:PRINT:NEXT
580 PRINT:PRINT:PRINT:PRINT:PRINT
590 PRINT" THIS IS MIRACLE TECHTEL"
600 PRINT
610 PRINT" ************************"
620 PRINT
630 PRINT" BULLETIN BOARD NO 1"
640 FOR T=1 TO 10500 :NEXT T
650 FOR P=1 TO 30:PRINT:NEXT
660 *FX 15,1
670 GOTO 160
```